
SPARQL QUERY LANGUAGE 
Primeri SPARQL upita



PRIMERI SPARQL UPITA NAD
LINKED MOVIE DATABASE

SPARQL endpoint: 
http://data.linkedmdb.org/sparql

Opis dataset-a:
http://linkedmdb.org/



Upit 1: Prikazati imena svih glumaca koji su igrali u filmovima sa datim
glumcem, npr. sa Kevin Bacon-om

PREFIX mov: <http://data.linkedmdb.org/resource/movie/>
SELECT DISTINCT ?actorName 
WHERE {
?kb  mov:actor_name "Kevin Bacon". 
?movie mov:actor ?kb;

mov:actor ?actor.
?actor mov:actor_name ?actorName.
FILTER (?kb != ?actor).

}



Upit 2: Prikazati imena svih glumaca koji su igrali bar u jednom filmu
koji je režirao Martin Scorsese i bar jednom filmu čiji je režiser Steven 
Spielberg

PREFIX mov: <http://data.linkedmdb.org/resource/movie/>
SELECT DISTINCT ?actorName 
WHERE {

?dir1   mov:director_name "Martin Scorsese".
?dir2   mov:director_name "Steven Spielberg".
?dir1movie mov:director ?dir1;

mov:actor ?actor.
?dir2movie mov:director ?dir2 ;

mov:actor ?actor .
?actor  mov:actor_name ?actorName.

}



Upit 3: Prikazati nazive (title) filmskih tema (subjects) za koje postoji
bar 50 filmovau LinkedMovie bazi znanja

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX mov: <http://data.linkedmdb.org/resource/movie/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

SELECT DISTINCT ?title (COUNT(?movie) AS ?count)
WHERE {

?subject  rdf:type mov:film_subject ; 
mov:film_subject_name ?title.

{ ?movie skos:subject ?subject . }
UNION

{ ?movie mov:film_subject ?subject . }
}
GROUP BY ?title
HAVING (COUNT(?movie) >= 50)
ORDER BY ?title



Upit 4: Prikazati imena svih režisera koji su režirali filmove čija je tema
‘mafia’; sortirati rezultate na osnovu imena režisera

PREFIX mov: <http://data.linkedmdb.org/resource/movie/>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

SELECT DISTINCT ?directorName 
WHERE {

{?movie skos:subject ?subject .} UNION  {?movie mov:film_subject ?subject .}
?subject mov:film_subject_name ?subjectLbl

FILTER (strStarts(lcase(?subjectLbl), “mafia”) ).
?movie mov:director ?director.
?director mov:director_name ?directorName .

}
ORDER BY ?directorName

U slučaju da SPARQL engine ne podržava SPARQL 1.1, ovu liniju zameniti sa: 
FILTER (regex(?subjectLbl, "^mafia", "i")) . 



PRIMERI SPARQL UPITA NAD
DBPEDIA BAZOM ZNANJA

SPARQL endpoint: http://dbpedia.org/sparql

SPARQL editors:
§ http://yasgui.laurensrietveld.nl/
§ http://dbpedia.org/snorql/



Upit 1: Prikazati imena i poreklo svih dobitnika Nobelove nagrade za
mir koji su rođeni u prvoj polovini XX veka; prikazati i po čemu su oni
poznati (dbo:knownFor), za one dobitnike za koje je taj podatak
raspoloživ.



PREFIX dbo: <http://dbpedia.org/ontology/> 
PREFIX dbr: <http://dbpedia.org/resource/> 
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#> 

SELECT DISTINCT ?name ?birthPlace ?deedLbl
WHERE {
?person  rdf:type  dbo:Person ;

rdfs:label ?name FILTER (lang(?name) = "en").
?person  dbo:award dbr:Nobel_Peace_Prize ;

dbo:birthPlace ?birthPlace ;
dbo:birthDate ?birthDate 
FILTER ((?birthDate > "1900-01-01"^^xsd:date) && 

(?birthDate < "1951-01-01"^^xsd:date)) .
OPTIONAL { ?person dbo:knownFor ?deed .

?deed rdfs:label ?deedLbl  FILTER (lang(?deedLbl) = "en") }  
}



Upit 2: Prikazati imena svih glumaca i glumica koji su dobili nagradu
Oskar za najbolju mušku, odnosno najbolju žensku ulogu, i koji su još
uvek živi. Sortirati rezultate prema starosti glumaca.



PREFIX dct: <http://purl.org/dc/terms/> 
PREFIX dbo: <http://dbpedia.org/ontology/>  
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT DISTINCT ?name
WHERE {
?person rdf:type dbo:Person ;

rdfs:label ?name FILTER (lang(?name) = "en").
?person dbo:birthDate ?birthDate .
{?person dct:subject dbr:Category:Best_Actor_Academy_Award_winners.}

UNION
{?person dct:subject dbr:Category:Best_Actress_Academy_Award_winners.} 
FILTER NOT EXISTS { ?person dbo:deathDate ?death. }

}
ORDER BY ?birthDate



Upit 3: Prikazati nazive prestonica (tj. glavnih gradova) koje pripadaju
CET vremenskoj zoni. Sortirati prikaz u opadajućem redosledu
veličineprestonica izražene kroz broj stanovnika. 



PREFIX dbo: <http://dbpedia.org/ontology/> 
PREFIX dbp: <http://dbpedia.org/property/> 
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT DISTINCT ?name 
WHERE {

?country rdf:type dbo:Country ;  dbo:capital ?city . 
FILTER NOT EXISTS { 

{?country dbo:dissolutionDate ?dateEnded .} UNION 
{?country dbo:dissolutionYear ?yearEnded .} }

?city rdfs:label ?name FILTER ( lang(?name) = "en" )
{?city dbo:timeZone dbr:Central_European_Time. }

UNION
{?city dbp:timezone ?timezone FILTER (str(?timezone) = "CET") }
?city dbp:populationTotal ?population .

}
ORDER BY DESC (?population)

Ovim filterom eliminišemo
iz skupa rezultata
nekadašnje prestonice, tj
prestonice država koje
više ne postoje



Upit 4: Prikazati nazive svih univerziteta koji su osnovani pre 
XX veka u zemljama koje pripadaju engleskom govornom
području. Takođe, prikazati naziv države u kojoj je univerzitet
lociran. Ako je raspoloživ podatak o broju studenata, prikazati
i taj podatak. Sortirati rezultate prema datumu osnivanja
univerziteta.



PREFIX dbo: <http://dbpedia.org/ontology/> 
PREFIX dbp: <http://dbpedia.org/property/> 
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT ?uniName ?cName ?students
WHERE {

?uni  rdf:type dbo:University ;
dbp:established ?creationDate ;
dbo:country ?country ;
rdfs:label ?uniName FILTER (lang(?uniName) = "en") .

?country dbo:language dbr:English_language ;
rdfs:label ?cName  FILTER (lang(?cName) = "en") .

OPTIONAL { ?uni dbo:numberOfUndergraduateStudents ?students . }
FILTER ( ?creationDate < "1900-01-01"^^xsd:date ) .  

}
ORDER BY (?creationDate)


